

Using a sub routine for incremental database extract in Qlik Sense

We find a useful feature in the script in Qlik is the ability to create sub routines. These can then be executed from other parts of the script. The sub routines can also have parameters passed into them.

The syntax for the sub routine is as follows

```
Sub <routinename> (parameter1, parameter2...)
```

Code to Execute

```
End Sub
```

The syntax to call a sub routine is as follows

```
Call <routinename> (parameter1, parameter2);
```

For an extraction process from a database which requires similar code to be accessed multiple times a sub routine is a way to reduce the amount of scripting required.

In the example below there is a sub routine that is connecting to a SQL server, three parameters are passed into it, vDb, the database, vOwn the object owner and vTable, the table to be extracted.

```
SUB ExtractData (vDb, vOwn, vTable)

LIB CONNECT TO '10.0.0.6 SQL Server';

SQL SELECT *
FROM $(vDb).$(vOwn).$(vTable);
|
ENDSUB

CALL ExtractData ('WideWorldImporters','Sales','Orders');
CALL ExtractData ('WideWorldImporters','Warehouse','StockItems');
```

The sub routine is then called twice for two separate tables, Orders and StockItems.

This can then be expanded further to store and drop the imported tables, as well as giving the ability to do an incremental load of a full load from the database.

```

SUB ExtractData(vDb, vOwn, vTable, vLoadType, vPK, vDateField)

LET vQvdFile = '$(vQvdLoc)$(vTable).qvd';

IF vLoadType = 'Inc' THEN

    // Set a variable indicating if the QVD exists or not. -1 is True, 0 is False
    LET vQvdExists = if(FileSize('$(vQvdFile)') > 0, -1, 0);

    IF vQvdExists = 0 THEN

        LET LastRun = TIMESTAMP(ADDYEARS(now(),-5),'YYYY-MM-DD hh:mm:ss');

        LET vWhereClause = 'WHERE ("$(vDateField)" <= ' & CHR(39) & '$(ThisRun)' & CHR(39) &
            ' AND "$(vDateField)" >= ' & CHR(39) & '$(LastRun)' & CHR(39) & ' )'

        ;

    ELSE

        LET vWhereClause = 'WHERE ("$(vDateField)" <= ' & CHR(39) & '$(ThisRun)' & CHR(39) &
            ' AND "$(vDateField)" >= ' & CHR(39) & '$(LastRun)' & CHR(39) & ' )'

        ;

    EndIF

    [$(vTable)]:
    LOAD $(vPK) AS KEY,
    *;
    SQL
    SELECT
    *
    FROM $(vDb).$(vOwn).$(vTable)
    $(vWhereClause);

    IF vQvdExists = -1 THEN

        CONCATENATE($(vTable))
        LOAD * FROM $(vQvdFile) (qvd)
        WHERE NOT EXISTS(KEY);

    EndIF

    STORE $(vTable) INTO $(vQvdFile);
    DROP TABLE $(vTable);

ELSEIF vLoadType = 'Full' THEN

    [$(vTable)]:
    SQL
    SELECT
    *
    FROM $(vDb).$(vOwn).$(vTable);

    STORE $(vTable) INTO $(vQvdFile);
    DROP TABLE $(vTable);

ENDIF

ENDSUB

```

Outside of the sub routine a couple of variables need to be declared as well as including the connection to the database machine.

The first of these variables is QvdLoc, this is the data connection where the QVD files will be stored, the second is the variable ThisRun which is the time of the current execution, this needs to be created in a data format that is accepted by the database. These variables need to be declared before the sub routine.

The sub routine is called by the code below, this also includes a variable being set recording the time of the current execution to be used next time the process runs.

```
CALL ExtractData('WideWorldImporters', 'Sales', 'OrderLines', 'Full','' ,'' );  
CALL ExtractData('WideWorldImporters', 'Sales', 'Orders', 'Inc','OrderID' ,'LastEditedWhen' );  
  
SET LastRun = ThisRun;
```

If another table needs to be extracted then an additional call can be added and the code will extract that table the next time the process runs.

It is also possible to use the function in multiple applications. The script can be saved as a text file on the Qlik server and then incorporated into different applications using the include command. This means that each database extraction in a Qlik installation would use a standard method. It would also reduce the development time if extraction from a new system is required.

For more information please contact us: support@informance.co.uk